
4 Attributierte Grammatiken

Der Parser prüft nur die syntaktische Korrektheit eines Quellprogramms, führt aber
noch keine Übersetzung durch. Diese erfolgt nun im Anschluss an die Syntaxanalyse.
Beim sogenannten Semantikanschluss fallen folgende Aufgaben an:

Prüfung von Kontextbedingungen. Hier werden Bedingungen überpüft, die

durch kontextfreie Grammatiken nicht beschrieben werden können, wie zum

Beispiel die Anforderung, dass jeder verwendete Name deklariert sein muss

oder dass die Typen von Operanden in Ausdrücken kompatibel sein müssen.

Dabei können weitere (semantische) Fehler entdeckt werden.

Verwaltung einer Symbolliste. Es wird ein Verzeichnis aller deklarierten

Namen samt ihren Eigenschaften wie Typ oder Adresse erstellt. Diese zentrale

Datenstruktur des Compilers nennt man die Symbolliste (siehe Kapitel 5). Sie

verwaltet auch die Struktur von Datentypen wie Arrays oder Klassen sowie die

Gültigkeitsbereiche von Namen.

Codeerzeugung. Wenn keine syntaktischen oder semantischen Fehler

entdeckt wurden, wird schließlich der Zielcode erzeugt. Dazu werden

Methoden aufgerufen, die Instruktionen der Zielmaschine erzeugen,

Sprungmarken verwalten oder dafür sorgen, dass für jede aufgerufene

Methode des Quellprogramms ein Bereich für ihre lokalen Variablen angelegt

wird.

In einem Einpass-Compiler (wie dem MicroJava-Compiler) werden diese Aktionen
direkt in den Parser integriert. Wir werden das in den Kapiteln 5 und 6 genauer
behandeln. Zur Beschreibung dieser Aktionen führen wir aber vorher eine Notation
ein, mit der sich Übersetzungsprozesse einfach und kompakt formulieren lassen,
nämlich attributierte Grammatiken.

Attributierte Grammatiken (ATGs) wurden von Donald Knuth eingeführt [Knut68].

In ihrer ursprünglichen Form gehen sie davon aus, dass aus dem Quellprogramm ein
Syntaxbaum aufgebaut wird, der dann ein- oder mehrmals durchlaufen wird, wobei
seine Knoten mit Eigenschaften wie Typen oder Adressen attributiert werden. Aus
diesem attributierten Syntaxbaum wird dann der Code erzeugt.

Wir verwenden hingegen eine einfachere Form von attributierten Grammatiken,
bei der semantische Aktionen schritthaltend mit der Syntaxanalyse ausgeführt
werden. Man nennt diese Form attributierter Grammatiken daher auch
syntaxgesteuerte Übersetzung (syntax-directed translation). Sie eignet sich besonders
für Einpass-Compiler.

4.1 Bestandteile
Wir sehen uns zuerst die Bestandteile attributierter Grammatiken an und zeigen
dann Beispiele für ihren Einsatz (auch außerhalb des engeren Compilerbaus). In
späteren Kapiteln dieses Buches (insbesondere bei der Codeerzeugung) werden wir
sie dann intensiv einsetzen, um damit die Übersetzung von MicroJava-Programmen
zu beschreiben.

Semantische Aktionen

Semantische Aktionen sind Codestücke (z.B. in Java), die in eine Grammatik an jener
Stelle eingefügt werden, an der sie beim Parsen ausgeführt werden sollen. Sehen wir
dazu ein Beispiel an:

Number = digit {digit}.

Eine Zahl besteht aus einer Ziffernfolge. Zahlen gehören zwar im Prinzip zur
lexikalischen Struktur einer Sprache, aber wir verwenden sie hier, wie wenn sie zur
Syntax der Sprache gehörten, um das Beispiel einfach zu erhalten.

Angenommen, die »Übersetzung« von Zahlen soll darin bestehen, die Ziffern
einer Zahl zu zählen und auszugeben. Das führt zu folgender Grammatik:

Number =

digit (. int n = 1; .)

{ digit (. n++; .)

} (. System.out.println(n); .)

.

Wie man sieht, werden semantische Aktionen als Codestücke zwischen den
Klammern (. und .) in die Grammatik eingefügt und werden vom Parser an der Stelle
ausgeführt, an der sie stehen. Nach der Erkennung der ersten Ziffer wird ein Zähler n

auf 1 gesetzt. Bei jedem weiteren Vorkommen einer Ziffer wird n erhöht und am Ende
der Ziffernfolge wird das Ergebnis ausgegeben.

Aus Lesbarkeitsgründen werden attributierte Grammatiken in tabellarischer
Form geschrieben: Links steht die Syntax und rechts daneben die semantischen
Aktionen, die nach den erkannten Syntaxsymbolen ausgeführt werden sollen. Die
»Übersetzung« von Zahlen liefert hier folgendes Ergebnis:

123 => 3

4711 => 4

9 => 1

Ausgangsattribute

Syntaxsymbole können Werte (Ausgangsattribute) liefern, die man dann in den
semantischen Aktionen verwenden kann. Solche Werte wirken wie
Ausgangsparameter der erkannten Syntaxsymbole. Die Erkennung einer Ziffer kann
zum Beispiel den Wert dieser Ziffer als Ausgangsattribut val liefern:

digit <↑val>

Attribute werden in spitze Klammern (< und >) gesetzt. Ihre Flussrichtung wird durch
einen Pfeil angedeutet. digit liefert also ein Ausgangsattribut val.

Wir können nun unsere Übersetzung etwas interessanter gestalten und anstatt
die Ziffern einer Zahl bloß zu zählen, den Wert der Zahl berechnen und ausgeben:

Number = (. int val, n; .)

digit <↑val>

{ digit <↑n> (. val = 10 * val + n; .)



9

} (. System.out.println(val); .)

.

Die erste Ziffer liefert einen Wert val. Jede weitere Ziffer liefert einen Wert n, der um
eine Zehnerpotenz verschoben zu val hinzuaddiert wird. Zum Schluss wird der Wert

der Zahl ausgegeben.

Eingangsattribute

Nonterminalsymbole können nicht nur Ausgangsattribute haben, sondern auch
Eingangsattribute, also Parameter, die ihnen von der Produktion, in der sie auftreten,
mitgegeben werden. Damit können wir Nonterminalsymbole parametrisieren. In
unserem Beispiel kann man durch ein Eingangsattribut base bestimmen, zu welcher
Basis (z.B. 10 oder 16) der Wert einer Zahl berechnet werden soll:

Number <↓base, ↑val> = (. int base, val, n; .)

digit <↑val>

{ digit <↑n> (. val = base * val + n; .)

}

.

Number hat ein Eingangsattribut base und ein Ausgangsattribut val. Die erste Ziffer
liefert wieder den Initialwert von val. Jede weitere Ziffer liefert einen Wert n, der mit

base * val + n in die Zahl aufgenommen wird. Das Ergebnis wird hier nicht ausgegeben,
sondern von Number als Ausgangsattribut val zurückgeliefert. Das

Übersetzungsergebnis lautet hier also:

base = 10, Ziffernfolge = 123 => val = 123

base = 16, Ziffernfolge = 123 => val = 291 (= 1*162 + 2*161 + 3*160)

Zusammenfassend ist eine attributierte Grammatik also eine übersichtliche Notation
zur Beschreibung von Übersetzungsprozessen. Sie besteht aus folgenden drei Teilen:

Produktionen in EBNF. Zum Beispiel:

IdentList = ident {"," ident}.

Attribute als Parameter von Syntaxsymbolen, zum Beispiel:

ident <↑name> Ausgangsattribute (liefern ein Übersetzungsergebnis)

IdentList <↓type> Eingangsattribute (stellen Kontext bereit)

Semantische Aktionen als beliebige Anweisungsfolgen (z.B. in Java):

(. beliebige Anweisungsfolge .)

4.2 Anwendungsbeispiele
Attributierte Grammatiken sind eine äußerst nützliche und kompakte Notation. Wir
werden sie in den späteren Kapiteln dieses Buches intensiv einsetzen, um die
Übersetzung von MicroJava-Programmen zu beschreiben. Sie sind aber auch
außerhalb des engeren Compilerbaus hilfreich, weshalb wir uns hier einige Beispiele
ansehen, die ihre Anwendung (auch außerhalb des Compilerbaus) zeigen.

Verarbeitung von Variablendeklarationen

Wir beginnen mit einem Beispiel aus dem MicroJava-Compiler, das die Verarbeitung
von Variablendeklarationen zeigt. Variablendeklarationen wie

int a, b, c;

beginnen mit einem Typ, auf den eine Liste von Namen folgt. Die Deklaration soll
bewirken, dass die deklarierten Namen samt ihrem Typ in die Symbolliste (Tab)

eingefügt werden. Die attributierte Grammatik dazu sieht wie folgt aus:

VarDecl (. Struct type; .)

= Type <↑type>

IdentList <↓type>

";" .

IdentList <↓type> (. Struct type; String name; .)

= ident <↑name> (. Tab.insert(name, type); .)

{ "," ident <↑name> (. Tab.insert(name, type); .)

} .

Type liefert eine Typbeschreibung (in Form eines Objekts type vom Typ Struct), die an
IdentList mitgegeben wird. In IdentList wird jeder deklarierte Name samt seinem Typ

mittels Tab.insert in die Symbolliste eingetragen.

Wie wird aber nun eine attributierte Grammatik im Compiler abgebildet? Dazu
integrieren wir die Attribute und semantischen Aktionen einfach in die
Parsermethoden. Attribute werden zu Parametern von Parsermethoden,
semantische Aktionen werden als Codestücke in die Parsermethoden eingefügt. Das
sieht dann wie folgt aus (Attribute und semantische Aktionen sind hier grau
dargestellt):

private static void VarDecl() {

Struct type;

type = Type();

IdentList(type);

check(semicolon);

}

private static void IdentList (Struct type) {

String name;

check(ident); name = t.val;

Tab.insert(name, type);

while (sym == comma) {

scan();

check(ident); name = t.val;

Tab.insert(name, type);

}

}

Eingangsattribute werden also zu Parametern, Ausgangsattribute zu
Rückgabewerten von Parsermethoden. Die Ausgangsattribute von
Terminalsymbolen werden aus dem zuletzt erkannten Token t entnommen, zum

Beispiel:

check(ident); name = t.val;

Wie man sieht, sind attributierte Grammatiken kompakter und lesbarer als
Parsermethoden. Die Syntax ist von der Semantik klarer getrennt. Die Darstellung
von Alternativen, Optionen und Iterationen ist kompakter als im Parsercode. Daher
werden wir in den nachfolgenden Kapiteln dieses Buches Übersetzungsprozesse
immer durch attributierte Grammatiken beschreiben.

Berechnung von Konstantenausdrücken

In diesem Beispiel wollen wir Konstantenausdrücke wie

3 * (2 + 4)

parsen und berechnen. Das kann wie folgt durch eine attributierte Grammatik
beschrieben werden:

Expr <↑val> (. int val, val1; .)

= Term <↑val>

{ "+" Term <↑val1> (. val = val + val1; .)

| "-" Term <↑val1> (. val = val - val1; .)

} .

Term <↑val> (. int val, val1; .)

= Factor <↑val>

{ "*" Factor <↑val1> (. val = val * val1; .)

| "/" Factor <↑val>1 (. val = val / val1; .)

} .

Factor <↑val> (. int val; .)

= number <↑val>

| "(" Expr <↑val> ")" .

Jeder Expr, Term und Factor liefert als Ausgangsattribut einen Wert, der dann

entsprechend dem Operator mit anderen Werten verknüpft wird. Abb. 4.1 zeigt einen
(gedachten) Syntaxbaum des Konstantenausdrucks, in dem die
Nonterminalsymbole mit den Werten ihrer Ausgangsattribute annotiert sind.

Abb. 4.1 Syntaxbaum eines Konstantenausdrucks (mit Ausgangsattributen)

Als Beispiel sehen wir uns nochmals an, wie attributierte Produktionen in
Parsermethoden umgesetzt werden und wählen dazu die Produktion von Expr:

private static int Expr() {

int val, val1;

val = Term();

for (;;) {

if (sym == plus) {

scan; val1 = Term(); val = val + val1;

} else if (sym == minus) {

scan(); val = Term(); val = val - val1;

} else break;

}

return val;

}

Beachten Sie, dass das Ausgangsattribut von Expr als Funktionswert zurückgegeben

wird. Expr muss daher als Funktionsmethode implementiert werden.

Verkaufsstatistik

Das nächste Beispiel stammt nicht aus dem engeren Compilerbau, sondern
beschreibt die Verarbeitung von Verkaufszahlen zu einer Verkaufsstatistik. Dennoch
können auch hier attributierte Grammatiken verwendet werden.

Nehmen wir an, wir haben eine Datei, in der Artikel (mit Artikelnummer) samt
ihren Verkaufsmengen gespeichert sind. Jede Artikelzeile ist durch einen Strichpunkt
abgeschlossen:

3451 2 5 3 7 ;

3452 4 5 1 ;

3453 1 1 ;

…

Wir wollen die Verkaufsmengen pro Artikel summieren und ausgeben, also:

3451 17

3452 10

2343 2

…

Die Eingabedatei hat eine syntaktische Struktur, nämlich:

File = {Article}.

Article = Code {Amount} ";".

Code = number.

Amount = number.

Immer wenn eine Eingabe syntaktisch strukturiert ist, können wir sie mit den
Techniken des Compilerbaus verarbeiten. Wir stellen dazu zunächst eine
attributierte Grammatik auf:

File (. int code, amount; .)

= { Article <↑code, ↑amount> (. System.out.println(code + " " + amount); .)

}.

Article <↑code, ↑amount> (. int code, x, amount = 0; .)

= number <↑code>

{ number <↑x> (. amount += x; .)

}

";".

Diese Grammatik können wir nun in Parsercode umsetzen. Article hat zwei
Ausgangsattribute. Da eine Parsermethode aber nur einen einzigen Funktionswert
zurückgeben kann, müssen wir die beiden Attribute zu einem Objekt einer
Hilfsklasse ArticleInfo zusammenfassen:

class ArticleInfo {

int code, amount;

}

Nun können wir die Parsermethoden implementieren (Attribute und semantische
Aktionen sind wieder grau dargestellt):

private static void File() {

while (sym == number) {

ArticleInfo a = Article();

System.out.println(a.code + " " + a.amount);

}

check(eof);

}

private static ArticleInfo Article() {

ArticleInfo a = new ArticleInfo();

check(number); a.code = t.numVal;

while (sym == number) {

scan(); int x = t.numVal;

a.amount += x;

}

check(semicolon);

return a;

}

Es fehlt nun nur noch ein Miniscanner, der die Terminalsymbole number, semicolon und

eof liefert. Dieser ist aber trivial zu implementieren.

Obwohl das Beispiel einfach ist, haben uns die Techniken des Compilerbaus
doch geholfen, es systematisch und kompakt zu implementieren. Eines der Ziele
dieses Buches ist es, Leser für solche Situationen zu sensibilisieren. Immer wenn
eine syntaktisch strukturierte Eingabe zu verarbeiten ist, können die Techniken des
Compilerbaus nutzbringend eingesetzt werden.

Bildbeschreibungssprache

Auch dieses Beispiel stammt nicht aus dem engeren Compilerbau, sondern
behandelt die Verarbeitung einer Bildbeschreibungssprache. Ein Bild, das in dieser
Sprache beschrieben ist, kann aus mehreren Figuren (Rechtecke, Kreise, Polygone
etc.) bestehen, die in einem bestimmten Format codiert sind. Konzentrieren wir uns
hier auf Polygone. Ein Polygon sei zum Beispiel wie folgt codiert:

POLY (10, 30) (50, 50) (40, 30) (50, 10) END

Es besteht aus mehreren Punkten, die jeweils durch eine x- und y-Koordinate

definiert sind. Polygone dieser Art sollen von einer Datei gelesen und auf dem
Bildschirm gezeichnet werden (siehe Abb. 4.2).

Abb. 4.2 Darstellung des Polygons

Zum Zeichnen verwenden wir die sogenannte »Turtle-Grafik«, bei der ein Stift (wie
eine Schildkröte) über die Zeichenfläche läuft und dabei eine Linie zieht. Wir

benutzen dazu die folgenden beiden Operationen:

Turtle.start(p); setzt den Stift auf den Punkt p

Turtle.move(q); bewegt den Stift zum Punkt q und zeichnet dabei eine Lin

Als geübte Compilerbauer beginnen wir nun damit, eine Grammatik für die
Beschreibung von Polygonen aufzustellen, die wie folgt aussieht:

Polygon = "POLY" Point {Point} "END".

Point = "(" number "," number ")".

Diese Grammatik können wir nun attributieren, wobei Point ein Ausgangsattribut des
Typs

class Pt { int x, y; }

liefern soll. Die attributierte Grammatik sieht folgendermaßen aus:

Polygon (. Pt p, q; .)

= "POLY"

Point <↑p> (. Turtle.start(p); .)

{ Point <↑q> (. Turtle.move(q); .)

}

"END" (. Turtle.move(p); .) .

Point <↑p> (. Pt p = new Pt(); .)

= "(" number <↑p.x>

"," number <↑p.y>

")" .

Der erste Punkt definiert den Startpunkt. Jeder weitere Punkt bewirkt, dass der Stift
zu dieser Position bewegt wird. Zum Schluss bewegen wir den Stift wieder zurück
zum Startpunkt, wodurch das Polygon geschlossen wird.

Ein Punkt ist durch zwei Zahlen definiert, die wir direkt in die Koordinaten p.x und

p.y übernehmen.

Wir brauchen nun diese attributierte Grammatik nur noch in Parsermethoden
umzusetzen und ähnliche Produktionen für die anderen Figurenarten (Rechtecke,
Kreise etc.) zu schreiben, dann ist die Verarbeitung unserer
Bildbeschreibungssprache fertig. Man sieht, dass auch hier die Techniken des
Compilerbaus nutzbringend eingesetzt werden konnten.

Umwandlung von Infix- in Postfix-Notation

Das letzte Beispiel, das wir uns anschauen, ist wieder näher am Compilerbau. Wir
wollen einen arithmetischen Ausdruck, der in Infix-Notation vorliegt, in einen
äquivalenten Ausdruck in Postfix-Notation umwandeln. Bei der Infix-Notation stehen
die Operatoren zwischen den Operanden, bei der Postfix-Notation nach den
Operanden, also:

Infix-Notation Postfix-Notation

3 + 4 * 2 3 4 2 * +

(3 + 4) * 2 3 4 + 2 *

Im ersten Postfix-Ausdruck werden beispielsweise die Operanden 4 und 2
multipliziert, bevor das Ergebnis zum Operanden 3 hinzuaddiert wird.

Diese Aufgabe erscheint kompliziert, ist aber mit etwas Compilerbau-Wissen sehr
einfach lösbar. Der Trick besteht darin, jeden Operanden sofort auszugeben und die
Ausgabe der Operatoren so lange hinauszuzögern, bis beide Operanden, auf die sich
der Operator bezieht, ausgegeben sind. Damit können wir die attributierte
Grammatik sofort formulieren:

Expr

= Term

{ "+" Term (. print("+ "); .)

| "-" Term (. print("- "); .)

}.

Term

= Factor

{ "*" Factor (. print("* "); .)

| "/" Factor (. print("/ "); .)

}.

Factor

= number <↑val> (. print(val + " "); .)

| "(" Expr ")".

Wenn ein Faktor eine Zahl ist, wird ihr Wert sofort ausgegeben. Bei Factor * Factor

werden die beiden Faktoren bereits in der Factor-Produktion ausgegeben und

anschließend der Operator "*" in der Term-Produktion. Ähnlich ist es bei Termen: Bei
Term + Term werden die beiden Terme in der Term-Produktion ausgegeben und

anschließend der Operator "+" in der Expr-Produktion. Wenn ein Faktor ein
geklammerter Ausdruck ist, wird dieser durch Aufruf der Expr-Produktion

ausgegeben, die wiederum Term und Faktor aufruft, um die Bestandteile des
Ausdrucks auszugeben.

Um die Arbeitsweise dieser attributierten Grammatik zu veranschaulichen, sehen
wir uns in Abb. 4.3 den (gedachten) Syntaxbaum des Ausdrucks

(3 + 4) * 2

an und annotieren ihn mit den Ausgaben der Elemente des Postfix-Ausdrucks.

Abb. 4.3 Umwandlung des Ausdrucks (3 + 4) * 2 in einen Postfix-Ausdruck

Expr ruft Term auf und dieser Factor. Factor erkennt die öffnende Klammer und ruft dann
Expr auf. Expr ruft wieder Term und Factor auf. Factor erkennt die Zahl 3 und gibt sie aus.

Dann kehren Factor und Term zurück. Expr erkennt den Operator "+", gibt ihn aber noch
nicht aus, sondern ruft wieder Term und Factor auf. Factor erkennt die Zahl 4 und gibt

sie aus. Dann kehren Factor und Term zurück. Die beiden Terme sind nun ausgegeben,
daher wird nun in Expr der verzögerte Operator "+" ausgegeben. Und so läuft die

Verarbeitung weiter. Die Ausgabe liefert den Postfix-Ausdruck 3 4 + 2 *.

4.3 Übungsaufgaben

1. Verstehen attributierter Grammatiken. Gegeben sei folgende attributierte

Grammatik:

N <↑n> (. int n; .)

= D <↑n>

{ D <↑x> (. n = 2 * n + x; .)

}.

D <↑x>

= "0" (. x = 0; .)

| "1" (. x = 1; .).

a) Was macht diese Grammatik, d.h., was liefert N als

Ausgangsattribut?

b) Übersetzen Sie die beiden Produktionen in Parsermethoden, die

auch die Attribute und semantischen Aktionen enthalten. Die

Terminalsymbole "0" und "1" können im Parser als zero und one

angesprochen werden.

2. Römische Zahlen. Gegeben sei folgende kontextfreie Grammatik für eine

vereinfachte Form römischer Zahlen zwischen 1 und 19 ("I" bedeutet den

Wert 1, "V" den Wert 5 und "X" den Wert 10):

RomanNumber = OneOrMore | FiveOrMore | TenOrMore.

OneOrMore = "I" ["I" ["I" ["I"]]].

FiveOrMore = "V" [OneOrMore].

TenOrMore = "X" [OneOrMore | FiveOrMore].

Erweitern Sie diese Grammatik zu einer attributierten Grammatik, sodass
RomanNumber als Ausgangsattribut den int-Wert der römischen Zahl liefert.

3. Rechnen mit Uhrzeiten. Die folgende kontextfreie Grammatik beschreibt

Ausdrücke, mit denen Uhrzeiten addiert und subtrahiert werden können:

TimeExpr = Time { "+" Time | "-" Time }.

Time = Hours ":" Minutes.

Hours = digit [digit].

Minutes = digit digit.

Erweitern Sie sie zu einer attributierten Grammatik, die als
Ausgangsattribut von TimeExpr die Anzahl der Minuten eines Zeitausdrucks
liefert, zum Beispiel:

3:24 + 1:00 => 264

21:15 + 0:30 - 12:00 => 585

Die Grammatik soll auch folgende Kontextbedingungen prüfen:

▪ Die Stunden müssen im Bereich 0..23 liegen.

▪ Die Minuten müssen im Bereich 0..59 liegen.

Sie können davon ausgehen, dass digit ein Ausgangsattribut val liefert.

4. Adressvergabe bei Variablendeklarationen. Gegeben sei folgende

kontextfreie Grammatik für Variablendeklarationen:

VarDeclList = VarDecl {VarDecl}.

VarDecl = Type ident {"," ident} ";".

Type = "boolean" | "char" | "int".

Erweitern Sie sie zu einer attributierten Grammatik, die für jede Variable
ihre Adresse sowie zum Schluss für alle Variablen den insgesamt benötigten
Speicherplatz ausgibt. Die Adressen beginnen bei 0. Variablen vom Typ
boolean benötigen 1 Byte, vom Typ char 2 Bytes und vom Typ int 4 Bytes. Für

boolean x, y;

int b;

char c;

soll zum Beispiel Folgendes ausgegeben werden:

x: 0

y: 1

b: 2

c: 6

Gesamtspeicherbedarf = 8

5. Kurse und Teilnehmer. Gegeben sei eine Textdatei, die Kurse und ihre

Teilnehmerinnen bzw. Teilnehmer beschreibt. Jeder Kurs hat einen Namen

und eine Liste von Teilnehmenden gemäß folgender Grammatik:

Courses = {Course}.

Course = "course" CourseName ":" Participants;

Participants = {Participant}.

CourseName = string.

Participant = string.

Attributieren Sie die Grammatik, sodass eine Teilnehmerliste

HashMap<String, ArrayList<String>> participants;

entsteht, in der für jeden Teilnehmer die Liste der von ihm besuchten Kurse
gespeichert wird. Geben Sie diese Liste anschließend aus. Entnehmen Sie
den Wert des Terminalsymbols string aus t.val.

6. Berechnung von Präfix-Ausdrücken. Arithmetische Ausdrücke können in

Präfix-Notation geschrieben werden, bei der der Operator vor den

Operanden steht (z.B. + a b). Hier sind einige Beispiele für Präfix-Ausdrücke

und ihre Bedeutung in der gewohnten Infix-Schreibweise:

+ 3 * 6 2 = 3 + (6 * 2)

* + 1 2 3 = (1 + 2) * 3

+ * 2 4 * 6 8 = (2 * 4) + (6 * 8)

a) Stellen Sie eine kontextfreie Grammatik für Präfix-Ausdrücke auf.

Ein Präfix-Ausdruck ist entweder eine Zahlenkonstante

(Terminalsymbol number) oder beginnt mit einem der Operatoren "+"

oder "*", auf den wieder zwei Präfix-Ausdrücke folgen. Verwenden

Sie Rekursion. Die Vorrangregeln zwischen "+" und "*" können

ignoriert werden, da sie implizit im Präfix-Ausdruck gegeben sind.

b) Attributieren Sie Ihre Grammatik so, dass der Wert eines Präfix-

Ausdrucks als Attribut geliefert wird. Den Wert von number können

Sie aus t.numVal entnehmen.

7. Skriptsprache für boolesche Ausdrücke. Folgende Grammatik beschreibt eine

kleine Sprache, mit der boolesche Ausdrücke berechnet, in Variablen

gespeichert und ausgegeben werden können:

Program = { Statement }.

Statement = ident "=" Expr ";"

| "print" Expr ";".

Expr = Term { "||" Term }.

Term = Factor { "&&" Factor }.

Factor = "true" | "false" | ident | "(" Expr ")" | "!" Factor.

Attributieren Sie diese Grammatik, sodass der Wert der booleschen
Ausdrücke berechnet und in Variablen abgespeichert oder ausgegeben
werden kann. Verwalten Sie die verwendeten Variablen und ihre Werte in
einer Hashtabelle (HashMap). Ein Beispielprogramm könnte wie folgt
aussehen:

big = true;

small = ! big;

ready = false;

print (big || small) && ready || big && ! ready;

8. Übersetzung von XML nach Json. Sowohl XML (Extensible Markup Language)
als auch Json (JavaScript Object Notation) sind Formate zur Beschreibung

markierter Datensätze. Ein XML-Datensatz, bestehend aus mehreren XML-

Elementen,

<person>

<name>Meier</name>

<city>Wien</city>

</person>

<person>

<name>Huber</name>

<city>Berlin</city>

</person>

wird in Json wie folgt dargestellt:

[

person: {

name: Meier,

city: Wien

},

person: {

name: Huber,

city: Berlin

}

]

Wir verwenden in diesem Beispiel eine vereinfachte XML-Syntax:

XML = { Element } .

Element = "<" ident ">"

( ident

| { Element }

)

"</" ident ">".

Attributieren Sie diese Grammatik, sodass ein XML-Datensatz in einen Json-
Datensatz übersetzt wird. Prüfen Sie auch, dass die Tag-Namen in <ident>

und </ident> übereinstimmen.

